

---

## *Stata 12/13 Tutorial 5*

**TOPIC:** Coefficient Differences and Dummy Variable Regressors

**DATA:** **auto1.dta** (a *Stata*-format dataset first created in *Stata 12/13 Tutorial 1*)

**TASKS:** *Stata 12/13 Tutorial 5* deals with using *dummy variable regressors* to allow and test for **coefficient differences** between the regression functions for foreign and domestic cars. It investigates examples of regression models that use dummy variable regressors to allow for ***intercept and/or slope coefficient differences*** between foreign and domestic cars. The emphasis throughout is on the interpretation of regression coefficients in models with dummy variable regressors and on the formulation of various tests for coefficient differences.

- The ***Stata* commands** that constitute the primary subject of this tutorial are:

<b>regress</b>	Used to perform OLS estimation of multiple linear regression models.
<b>scalar</b>	Used to save as scalars elements of the coefficient vector and the coefficient covariance matrix after OLS estimation.
<b>scalar list</b>	Used to display the values of previously-defined scalars.
<b>matrix</b>	Used to save the vector of OLS coefficient estimates.
<b>matrix list</b>	Used to display the vector of OLS coefficient estimates.
<b>lincom</b>	Used after estimation to compute linear combinations of coefficient estimates and associated test statistics.
<b>test</b>	Used to compute Wald F-tests of linear coefficient restrictions, with <b>notest</b> and <b>accumulate</b> options.
<b>return list</b>	Used to display all temporarily saved results from the most recent <b>test</b> or <b>lincom</b> command.

- No ***Stata* statistical functions** are used in this tutorial.

**NOTE:** *Stata* commands are *case sensitive*. All *Stata* command names must be typed in the Command window in **lower case letters**.

---

## □ **Preparing for Your *Stata* Session**

---

Before beginning your *Stata* session, use Windows Explorer to copy the *Stata*-format dataset **auto1.dta** to the *Stata working directory* on the C:-drive or D:-drive of the computer at which you are working. If you did not save the *Stata*-format dataset **auto1.dta** while doing *Stata 12/13 Tutorial 1*, you will have to re-create the text-format dataset **auto1.raw** as a *Stata*-format dataset using the commands outlined in *Stata 12/13 Tutorial 1*.

- **On the computers in Dunning 350**, the default *Stata* working directory is usually **C:\data**.
- **On the computers in MC B111**, the default *Stata* working directory is usually **D:\courses**.

---

## □ **Start Your *Stata* Session**

---

**To start your *Stata* session**, double-click on the ***Stata* icon** on the Windows desktop or in the **Start** menu under **Programs**.

After you double-click the ***Stata* icon**, you will see the familiar screen of five *Stata* windows.

---

## □ **Record Your *Stata* Session – log using**

---

**To record your *Stata* session**, including all the *Stata* commands you enter and the results (output) produced by these commands, make a text-format **.log** file named **452tutorial5.log**. To open (begin) the log file **452tutorial5.log**, enter in the Command window:

```
log using 452tutorial5.log
```

This command opens a plain text-format (ASCII) file called **452tutorial5.log** in the current *Stata* working directory.

**Note:** It is important to include the **.log** file extension when opening a log file; if you do not, your log file will be in smcl format, a format that only *Stata* can read. Once

you have opened the **452tutorial5.log** file, a copy of all the commands you enter during your *Stata* session and of all the results they produce is recorded in that **452tutorial5.log** file.

#### □ **Record Only Your *Stata* Commands – cmdlog using**

---

**To record only the *Stata* commands you type during your *Stata* session**, you can use the *Stata* **cmdlog using** command. To start (open) the command log file **452tutorial5.txt**, enter in the Command window:

```
cmdlog using 452tutorial5
```

This command opens a plain text-format (ASCII) file called **452tutorial5.txt** in the current *Stata* working directory. All commands you enter during your *Stata* session are recorded in this file.

#### □ **Loading a *Stata*-Format Dataset into *Stata* – use**

---

**To check that the *Stata*-format dataset ‘auto1.dta’ is in the current *Stata* working directory** of the computer at which you are working, type in the Command window:

```
dir auto1.*
```

You should see in the *Stata* Results window the filename ‘auto1.dta’.

**To load, or read, into memory the *Stata*-format dataset auto1.dta**, type in the Command window:

```
use auto1
```

This command loads into memory the *Stata*-format dataset **auto1.dta**.

**To summarize the contents of the current dataset**, use the **describe** and **summarize** commands. Type in the Command window the following commands:

```
describe  
summarize
```

---

## □ **Model 1 -- Identical Regression Coefficients for Domestic and Foreign Cars**

---

The dataset **auto1.dta** contains a **binary indicator (dummy) variable**  $foreign_i$  that distinguishes between two categories of cars sold in North America: **domestic cars**, defined as cars that are manufactured in North America; and **foreign cars**, defined as cars that are manufactured outside North America. The **foreign-car indicator, or dummy, variable** is defined as follows:

$foreign_i = 1$  if the  $i$ -th car is a foreign car  
 $= 0$  if the  $i$ -th car is a domestic car

- To see for yourself how the dummy variable  $foreign_i$  is coded, as well as how many cars in the sample are domestic cars and how many are foreign cars, enter the following commands:

```
codebook foreign
tab1 foreign
summarize foreign, detail
```

In this section, we estimate by OLS a regression model for North American car prices that constrains all the regression coefficients to be the same for domestic cars and foreign cars. Write the **population regression equation for Model 1** as:

$$price_i = \beta_0 + \beta_1 wgt_i + \beta_2 mpg_i + \beta_3 wgt_i^2 + \beta_4 mpg_i^2 + \beta_5 wgt_i mpg_i + u_i \quad (1)$$

Since regression equation (1) makes no distinction whatsoever between domestic and foreign cars, it implicitly restricts all the regression coefficients  $\beta_j$  ( $j = 0, 1, \dots, 5$ ) to be equal or identical for domestic and foreign cars.

- First, if they are not already included in your current dataset, you will have to generate the regressors  $wgt_i^2$ ,  $mpg_i^2$  and  $wgt_i mpg_i$  that enter Model 2. In that case, you should enter the following commands:

```
generate wgtsq = wgt^2
generate mpgsq = mpg^2
generate wgtmpg = wgt*mpg
summarize price wgt mpg wgtsq mpgsq wgtmpg
```

- Estimate by OLS regression equation (1) on the full sample of 74 observations for both foreign and domestic cars. Enter the **regress** command:

```
regress price wgt mpg wgtsq mpgsq wgtmpg
```

You will want to refer back to the OLS estimates of Model 1 produced by this **regress** command, as Model 1 represents the benchmark for all subsequent models in this tutorial.

## □ Model 2 – Different Intercepts for Domestic and Foreign Cars

---

In this section, we consider two different but equivalent regression models that allow only the intercept coefficient in Model 1 to differ between foreign and domestic cars. Each model adds to Model 1 an additive dummy variable regressor that distinguishes between cars manufactured in North America -- called domestic cars -- and cars manufactured outside North America -- called foreign cars.

The dataset **auto1.dta** contains a *foreign-car indicator variable* named *foreign<sub>i</sub>* that is defined as follows:

$foreign_i = 1$  if the *i*-th car is a foreign car  
 $= 0$  if the *i*-th car is a domestic car

An alternative way to distinguish between foreign and domestic cars is to define a *domestic-car indicator variable* named *domestic<sub>i</sub>*; it is defined as follows:

$domestic_i = 1$  if the *i*-th car is a domestic car  
 $= 0$  if the *i*-th car is a foreign car

Given their definitions, it follows that the indicator variables *foreign<sub>i</sub>* and *domestic<sub>i</sub>* must satisfy the following **adding-up property** for each sample observation:

$foreign_i + domestic_i = 1$  for all *i*.

This adding-up property implies that each indicator variable can be expressed in terms of the other:

$\text{foreign}_i = 1 - \text{domestic}_i$  for all  $i$ .

$\text{domestic}_i = 1 - \text{foreign}_i$  for all  $i$ .

Either indicator variable can be used to distinguish between domestic and foreign cars. In the remainder of this section, you explore the relationship between **two different but equivalent models** that allow the intercept coefficient to differ between foreign and domestic cars.

- First, you will need to create the domestic-car indicator variable *domestic<sub>i</sub>*. Since the foreign-car indicator variable *foreign<sub>i</sub>* already exists in your dataset, the easiest way to create the domestic-car indicator variable is to enter the following **generate** command:

```
generate domestic = 1 - foreign
```

- You can verify the definitional relationship between these two indicator variables by inspecting the results of the following commands:

```
summarize domestic foreign
tab2 domestic foreign
```

### **Model 2.1 – Version 1 of Model 2: Base group is domestic cars**

Model 2.1 is obtained by adding to Model 1 the foreign-car dummy variable *foreign<sub>i</sub>* as an additional additive regressor.

The **population regression equation for Model 2.1** is:

$$\text{price}_i = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \delta_0 \text{foreign}_i + u_i \quad \dots (2.1)$$

- The **population regression function for Model 2.1** is obtained by taking the conditional expectation of regression equation (2.1) for any given values of the three explanatory variables  $\text{wgt}_i$ ,  $\text{mpg}_i$ , and  $\text{foreign}_i$ :

$$E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{foreign}_i)$$

$$= \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \delta_0 \text{foreign}_i \quad (2.1')$$

- The **domestic-car population regression function** implied by Model 2.1 is obtained by setting the *foreign-car* indicator variable  $\text{foreign}_i = 0$  in (2.1'):

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{foreign}_i = 0) \\ = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

- The **foreign-car population regression function** implied by Model 2.1 is obtained by setting the *foreign-car* indicator variable  $\text{foreign}_i = 1$  in (2.1'):

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{foreign}_i = 1) \\ = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \delta_0 \\ = (\beta_0 + \delta_0) + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

### ***Interpretation of coefficients in Model 2.1***

- ♦ The **slope coefficients**  $\beta_1, \beta_2, \beta_3, \beta_4,$  and  $\beta_5$  are the same for both foreign and domestic cars.
- ♦ The **intercept coefficient for domestic cars** in Model 2.1  $= \beta_0$ .
- ♦ The **intercept coefficient for foreign cars** in Model 2.1  $= \beta_0 + \delta_0$ .
- ♦ The **slope coefficient on the foreign-car dummy variable** in Model 2.1 is the **foreign-domestic car difference** in intercept coefficients; specifically,  $\delta_0$  = the intercept coefficient for foreign cars ( $\beta_0 + \delta_0$ ) – the intercept coefficient for domestic cars ( $\beta_0$ ).
- Estimate Model 2.1 by OLS on the full sample of domestic and foreign cars. Enter the **regress** command:

```
regress price wgt mpg wgtsq mpgsq wgtmpg foreign
```

Interpret the coefficient estimate of the  $\text{foreign}_i$  indicator variable produced by this **regress** command. Would you conclude from the results that the mean prices of foreign and domestic cars of the *same* size and fuel efficiency (*same* values of  $\text{wgt}_i$  and  $\text{mpg}_i$ ) are similar or different?

- Use the **lincom** command to compute and test the *intercept coefficient estimate for foreign cars*. Enter either of the following commands:

```
lincom _cons + foreign
lincom _b[_cons] + _b[foreign]
```

### Model 2.2 – Version 2 of Model 2: Base group is foreign cars

Model 2.2 is obtained by adding to Model 1 the domestic-car dummy variable *domestic<sub>i</sub>* as an additional additive regressor.

The **population regression equation for Model 2.2** is:

$$\text{price}_i = \alpha_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \gamma_0 \text{domestic}_i + u_i \quad \dots (2.2)$$

- ♦ The **population regression function for Model 2.2** is obtained by taking the conditional expectation of regression equation (2.2) for any given values of the three explanatory variables *wgt<sub>i</sub>*, *mpg<sub>i</sub>*, and *domestic<sub>i</sub>*:

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{domestic}_i) \\ = \alpha_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \gamma_0 \text{domestic}_i \quad (2.2') \end{aligned}$$

- ♦ The **domestic-car population regression function** implied by Model 2.2 is obtained by setting the *domestic-car* indicator variable *domestic<sub>i</sub>* = 1 in (2.2'):

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{domestic}_i = 1) \\ = \alpha_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \gamma_0 \\ = (\alpha_0 + \gamma_0) + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

- ♦ The **foreign-car population regression function** implied by Model 2.2 is obtained by setting the *domestic-car* indicator variable *domestic<sub>i</sub>* = 0 in (2.2'):

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, \text{domestic}_i = 0) \\ = \alpha_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

***Interpretation of coefficients in Model 2.2***

- ◆ The **slope coefficients**  $\beta_1, \beta_2, \beta_3, \beta_4,$  and  $\beta_5$  are the same for both foreign and domestic cars.
  - ◆ The **intercept coefficient for foreign cars** in Model 2.2 =  $\alpha_0$ .
  - ◆ The **intercept coefficient for domestic cars** in Model 2.2 =  $\alpha_0 + \gamma_0$ .
  - ◆ The **slope coefficient on the domestic-car dummy variable** in Model 2.2 is the **domestic-foreign car difference** in intercept coefficients; specifically,  $\gamma_0$  = the intercept coefficient for domestic cars ( $\alpha_0 + \gamma_0$ ) – the intercept coefficient for foreign cars ( $\alpha_0$ ).
- Estimate Model 2.2 by OLS on the full sample of domestic and foreign cars. Enter the **regress** command:

```
regress price wgt mpg wgtsq mpgsq wgtmpg domestic
```

Interpret the coefficient estimate of the domestic<sub>i</sub> indicator variable produced by this **regress** command. Would you conclude from the results of this **regress** command that the mean prices of foreign and domestic cars of the *same* size and fuel efficiency (*same* values of wgt<sub>i</sub> and mpg<sub>i</sub>) are similar or different? Is your conclusion identical to that implied by the coefficient estimate of the foreign<sub>i</sub> indicator variable in Model 2.1?

- Use the **lincom** command to compute and test the **intercept coefficient estimate for domestic cars**. Enter either of the commands:

```
lincom _cons + domestic
lincom _b[_cons] + _b[domestic]
```

Compare the **intercept coefficient estimate for domestic cars** given by this **lincom** command with the intercept coefficient estimate for domestic cars given by the OLS sample regression equation for Model 2.1.

Compare the **foreign-car intercept coefficient estimate** from Model 2.2 with that implied by Model 2.1.

---

## □ Estimating Separate Regression Equations for Domestic and Foreign Cars

---

This section demonstrates how to compute separate OLS estimates of the regression equations for domestic and foreign cars. Because these regression equations are estimated separately on the subsamples of observations for domestic and foreign cars, they allow *all* of the regression coefficients in Model 1 – both the intercept and slope coefficients – to take different values for domestic and foreign cars.

Write the population regression equation for *domestic car prices* as:

$$\text{price}_i = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + u_i \quad \begin{array}{l} \text{domestic}_i = 1 \\ \text{foreign}_i = 0 \end{array}$$

Write the population regression equation for *foreign car prices* as:

$$\text{price}_i = \alpha_0 + \alpha_1 \text{wgt}_i + \alpha_2 \text{mpg}_i + \alpha_3 \text{wgt}_i^2 + \alpha_4 \text{mpg}_i^2 + \alpha_5 \text{wgt}_i \text{mpg}_i + u_i \quad \begin{array}{l} \text{foreign}_i = 1 \\ \text{domestic}_i = 0 \end{array}$$

- Compute **OLS estimates of the regression equation for *domestic car prices*** on the subsample of observations for which  $\text{domestic}_i = 1$ . Enter the following **regress** command with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if domestic==1
```

- Compute **OLS estimates of the regression equation for *domestic car prices*** on the subsample of observations for which  $\text{foreign}_i = 0$ . Enter the following **regress** command with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if foreign==0
```

This command gives the same results as the previous **regress** command because, by definition,  $\text{domestic}_i = 1$  is equivalent to  $\text{foreign}_i = 0$ .

- Save the **vector of OLS coefficient estimates for *domestic cars***. Enter the following **matrix** and **matrix list** commands:

```
matrix bvec = e(b)
matrix list bvec
```

- Save as scalars the **OLS coefficient estimates for *domestic cars***. Enter the following **scalar** and **scalar list** commands:

```
scalar b1 = _b[wgt]
scalar b2 = _b[mpg]
scalar b3 = _b[wgtsq]
scalar b4 = _b[mpgsq]
scalar b5 = _b[wgtmpg]
scalar b0 = _b[_cons]
scalar list b1 b2 b3 b4 b5 b0
```

- Compute **OLS estimates of the regression equation for *foreign car prices*** on the subsample of observations for which  $\text{domestic}_i = 0$ . Enter the following **regress** command with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if domestic==0
```

- Compute **OLS estimates of the regression equation for *foreign car prices*** on the subsample of observations for which  $\text{foreign}_i = 1$ . Enter the following **regress** command with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if foreign==1
```

This command gives the same results as the previous **regress** command because, by definition,  $\text{domestic}_i = 0$  is equivalent to  $\text{foreign}_i = 1$ .

- Save the **vector of OLS coefficient estimates for *foreign cars***. Enter the following **matrix** and **matrix list** commands:

```
matrix avec = e(b)
matrix list avec
```

- Save as scalars the **OLS coefficient estimates for *foreign cars***. Enter the following **scalar** and **scalar list** commands:

```
scalar a1 = _b[wgt]
scalar a2 = _b[mpg]
scalar a3 = _b[wgtsq]
scalar a4 = _b[mpgsq]
scalar a5 = _b[wgtmpg]
scalar a0 = _b[_cons]
scalar list a1 a2 a3 a4 a5 a0
```

---

### □ **Model 3 – A Full-Interaction Regression Model: Alternative Specifications**

---

To allow all the regression coefficients in Model 1 to differ between domestic and foreign cars, we can use either the foreign-car dummy variable  $f_i = \text{foreign}_i$  or the domestic-car dummy variable  $d_i = \text{domestic}_i$  to formulate a full-interaction regression model.

Before specifying these full-interaction regression models, it will be convenient to give shorter variable names to the foreign-car dummy variable  $\text{foreign}_i$  and the domestic-car dummy variable  $\text{domestic}_i$ . In particular, denote the foreign-car dummy variable  $\text{foreign}_i$  as  $f_i$  and the domestic-car dummy variable  $\text{domestic}_i$  as  $d_i$ . This will save some typing effort in entering commands, and will also permit a more compact representation of the alternative specifications of the full-interaction regression model.

- Enter the following **rename** commands:

```
rename foreign f
rename domestic d
```

Either the foreign-car dummy variable  $f_i$  or the domestic-car dummy variable  $d_i$  can be used to formulate the full-interaction specification of Model 1 that allows all of the regression coefficients in Model 1 to take different values in the foreign-car and domestic-car regression functions. There are therefore two different specifications of the full-interaction model; these are labeled Model 3.1 and Model 3.2.

#### ***Model 3.1 -- Version 1 of Model 3: Base group is domestic cars***

$$Y_i = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i + \delta_0 f_i + \delta_1 f_i \text{wgt}_i + \delta_2 f_i \text{mpg}_i + \delta_3 f_i \text{wgt}_i^2 + \delta_4 f_i \text{mpg}_i^2 + \delta_5 f_i \text{wgt}_i \text{mpg}_i + u_i \quad (3.1)$$

#### ***Model 3.2 -- Version 2 of Model 3: Base group is foreign cars***

$$Y_i = \alpha_0 + \alpha_1 \text{wgt}_i + \alpha_2 \text{mpg}_i + \alpha_3 \text{wgt}_i^2 + \alpha_4 \text{mpg}_i^2 + \alpha_5 \text{wgt}_i \text{mpg}_i + \gamma_0 d_i + \gamma_1 d_i \text{wgt}_i + \gamma_2 d_i \text{mpg}_i + \gamma_3 d_i \text{wgt}_i^2 + \gamma_4 d_i \text{mpg}_i^2 + \gamma_5 d_i \text{wgt}_i \text{mpg}_i + u_i \quad (3.2)$$

***Interpreting the Coefficients in Model 3.1***

- ♦ The **regression coefficients**  $\beta_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.1) are the regression coefficients for ***domestic cars***.

$\beta_j = \text{domestic-car}$  regression coefficient for regressor  $j$ .

To see why, use equation (3.1) to take the conditional mean of  $\text{price}_i$  for given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  when  $f_i = 0$ ; this gives you the regression function for domestic cars, which are the base group in equation (3.1).

- ♦ The **regression coefficients**  $\delta_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.1) are the ***foreign-domestic coefficient differences***

$\delta_j = \alpha_j - \beta_j = \text{foreign-car}$  coefficient – ***domestic-car*** coefficient.

This implies that the regression coefficients for ***foreign cars*** in Model 3.1 equal

$$\begin{aligned} \alpha_j &= \beta_j + \delta_j \quad (= \beta_j + \alpha_j - \beta_j) \\ &= \text{domestic-car} \text{ coefficient} + \text{foreign-domestic} \text{ coefficient difference.} \end{aligned}$$
***Interpreting the Coefficients in Model 3.2***

- ♦ The **regression coefficients**  $\alpha_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.2) are the regression coefficients for ***foreign cars***.

$\alpha_j = \text{foreign-car}$  regression coefficient for regressor  $j$ .

To see why, use equation (3.2) to take the conditional mean of  $\text{price}_i$  for given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  when  $d_i = 0$ ; this gives you the regression function for foreign cars, which are the base group in equation (3.2).

- ♦ The **regression coefficients**  $\gamma_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.2) are the ***domestic-foreign coefficient differences***

$\gamma_j = \beta_j - \alpha_j = \text{domestic-car}$  coefficient – ***foreign-car*** coefficient.

This implies that the regression coefficients for *domestic cars* in Model 3.2 equal

$$\begin{aligned}\beta_j &= \alpha_j + \gamma_j (= \alpha_j + \beta_j - \alpha_j) \\ &= \textit{foreign-car coefficient} + \textit{domestic-foreign coefficient difference}.\end{aligned}$$

### ***How are the Regression Coefficients in Models 3.1 and 3.2 Related?***

The regression coefficients  $\delta_j$  in equation (3.1) and  $\gamma_j$  in equation (3.2) are related by definition. Since

$$\delta_j = \alpha_j - \beta_j = \textit{foreign-car coefficient} - \textit{domestic-car coefficient}$$

and

$$\gamma_j = \beta_j - \alpha_j = \textit{domestic-car coefficient} - \textit{foreign-car coefficient},$$

it follows that

$$\delta_j = -\gamma_j \quad \text{or equivalently that} \quad \gamma_j = -\delta_j.$$

In other words, the regression coefficients  $\delta_j$  in Model 3.1 are ***equal in magnitude but opposite in sign*** to the regression coefficients  $\gamma_j$  in equation (3.2).

In the remainder of this tutorial, you will confirm the interpretation of the regression coefficients in Models 3.1 and 3.2 and the relationships between them by estimating regression equations (3.1) and (3.2) and comparing their OLS coefficient estimates.

### **Model 3.1 – Version 1 of Model 3**

$$\begin{aligned}Y_i &= \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \\ &\quad + \delta_0 f_i + \delta_1 f_i \text{wgt}_i + \delta_2 f_i \text{mpg}_i + \delta_3 f_i \text{wgt}_i^2 + \delta_4 f_i \text{mpg}_i^2 + \delta_5 f_i \text{wgt}_i \text{mpg}_i + u_i\end{aligned}\tag{3.1}$$

- Before estimating Model 3.1, it is necessary to create the interaction terms in the foreign-car indicator variable  $f_i$ , namely  $f_i \text{wgt}_i$ ,  $f_i \text{mpg}_i$ ,  $f_i \text{wgt}_i^2$ ,  $f_i \text{mpg}_i^2$  and  $f_i \text{wgt}_i \text{mpg}_i$ . Enter the following **generate** commands:

```

generate fwgt = f*wgt
generate fmpg = f*mpg
generate fwgtsq = f*wgtsq
generate fmpgsq = f*mpgsq
generate fwgtmpg = f*wgtmpg

```

- Estimate full-interaction regression equation (3.1) by OLS. Enter *on one line* the **regress** command:

```

regress price wgt mpg wgtsq mpgsq wgtmpg f fwgt fmpg
fwgtsq fmpgsq fwgtmpg

```

- Use the **scalar** command to save the OLS coefficient estimates of equation (3.1), and the **scalar list** command to display the saved scalars. Enter the commands:

```

scalar beta1 = _b[wgt]
scalar beta2 = _b[mpg]
scalar beta3 = _b[wgtsq]
scalar beta4 = _b[mpgsq]
scalar beta5 = _b[wgtmpg]
scalar beta0 = _b[_cons]
scalar delta1 = _b[fwgt]
scalar delta2 = _b[fmpg]
scalar delta3 = _b[fwgtsq]
scalar delta4 = _b[fmpgsq]
scalar delta5 = _b[fwgtmpg]
scalar delta0 = _b[f]
scalar list beta1 beta2 beta3 beta4 beta5 beta0
scalar list delta1 delta2 delta3 delta4 delta5 delta0

```

- Use the **lincom** command to compute the *foreign-car coefficient estimates* implied by the OLS coefficient estimates of equation (3.1), and to perform two-tail significance tests of these coefficient estimates. Enter the commands:

```

lincom _b[wgt] + _b[fwgt]
lincom _b[mpg] + _b[fmpg]
lincom _b[wgtsq] + _b[fwgtsq]
lincom _b[mpgsq] + _b[fmpgsq]
lincom _b[wgtmpg] + _b[fwgtmpg]
lincom _b[_cons] + _b[f]

```

Which of the foreign-car coefficient estimates are statistically significant at the 1 percent significance level, at the 5 percent significance level, and at the 10 percent significance level?

**Model 3.1: Tests for Coefficient Differences Between Foreign and Domestic Cars**

- ◆ **Test 1 - Model 3.1:** Test the proposition that **domestic cars and foreign cars have identical mean prices** for any given values of  $wgt_i$  and  $mpg_i$ .
- ◆ The foreign-domestic difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.1 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, f_i = 1) - E(\text{price}_i | wgt_i, mpg_i, f_i = 0) \\ = \delta_0 + \delta_1 wgt_i + \delta_2 mpg_i + \delta_3 wgt_i^2 + \delta_4 mpg_i^2 + \delta_5 wgt_i mpg_i \end{aligned}$$

- ◆ The proposition to be tested is that

$$E(\text{price}_i | wgt_i, mpg_i, f_i = 1) - E(\text{price}_i | wgt_i, mpg_i, f_i = 0) = 0 \quad \forall i$$

and hence that

$$\delta_0 + \delta_1 wgt_i + \delta_2 mpg_i + \delta_3 wgt_i^2 + \delta_4 mpg_i^2 + \delta_5 wgt_i mpg_i = 0 \quad \text{for all } i$$

A **sufficient condition** for the above equation to be true is that **all six of the  $\delta_j$  coefficients** in Model 3.1 jointly **equal zero**.

- ◆ The **null and alternative hypotheses** are as follows:

$$H_0: \delta_j = 0 \quad \text{for all } j = 0, 1, \dots, 5$$

$$\delta_0 = 0 \text{ and } \delta_1 = 0 \text{ and } \delta_2 = 0 \text{ and } \delta_3 = 0 \text{ and } \delta_4 = 0 \text{ and } \delta_5 = 0$$

$$H_1: \delta_j \neq 0 \quad j = 0, 1, \dots, 5$$

$$\delta_0 \neq 0 \text{ and/or } \delta_1 \neq 0 \text{ and/or } \delta_2 \neq 0 \text{ and/or } \delta_3 \neq 0 \text{ and/or } \delta_4 \neq 0 \text{ and/or } \delta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test f fwgt fmpg fwgtsq fmpgsq fwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

- ◆ **Test 2 - Model 3.1:** Test the proposition that the **foreign-domestic difference in mean car prices is a constant**, i.e., that it does not depend on the explanatory variables  $wgt_i$  and  $mpg_i$ .
- ◆ Recall that the foreign-domestic difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.1 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, f_i = 1) - E(\text{price}_i | wgt_i, mpg_i, f_i = 0) \\ = \delta_0 + \delta_1 wgt_i + \delta_2 mpg_i + \delta_3 wgt_i^2 + \delta_4 mpg_i^2 + \delta_5 wgt_i mpg_i \end{aligned}$$

- ◆ The proposition to be tested is that

$$\delta_0 + \delta_1 wgt_i + \delta_2 mpg_i + \delta_3 wgt_i^2 + \delta_4 mpg_i^2 + \delta_5 wgt_i mpg_i = \delta_0 \quad \text{for all } i$$

A **sufficient condition** for the above equation to be true is that **the five  $\delta_j$  coefficients** on the foreign dummy variable interaction terms in Model 3.1 all **equal zero**.

- ◆ The **null and alternative hypotheses** are as follows:

$$H_0: \delta_j = 0 \quad \text{for all } j = 1, \dots, 5$$

$$\delta_1 = 0 \text{ and } \delta_2 = 0 \text{ and } \delta_3 = 0 \text{ and } \delta_4 = 0 \text{ and } \delta_5 = 0$$

$$H_1: \delta_j \neq 0 \quad j = 1, \dots, 5$$

$$\delta_1 \neq 0 \text{ and/or } \delta_2 \neq 0 \text{ and/or } \delta_3 \neq 0 \text{ and/or } \delta_4 \neq 0 \text{ and/or } \delta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test fwgt fmpg fwgtsq fmpgsq fwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

- ◆ ***Test 3 - Model 3.1:*** Test the proposition that the *foreign-domestic difference in mean car prices does not depend on the explanatory variable wgt<sub>i</sub>*.
- ◆ This proposition is empirically equivalent to the following three statements:
  - (1) The relationship of price<sub>i</sub> to wgt<sub>i</sub> is identical for domestic and foreign cars.
  - (2) The marginal effect of wgt<sub>i</sub> on price<sub>i</sub> is identical for domestic and foreign cars.
  - (3) The foreign-domestic difference in mean car prices is a function only of the explanatory variable mpg<sub>i</sub>.
- ◆ Recall that the foreign-domestic difference in conditional mean car prices for any specified values of wgt<sub>i</sub> and mpg<sub>i</sub> is given in Model 3.1 by

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0) \\ = \delta_0 + \delta_1 \text{wgt}_i + \delta_2 \text{mpg}_i + \delta_3 \text{wgt}_i^2 + \delta_4 \text{mpg}_i^2 + \delta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

The foreign-domestic difference in mean car prices does not depend on wgt<sub>i</sub> if and only if  $\delta_1 = 0$  and  $\delta_3 = 0$  and  $\delta_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0) \\ = \delta_0 + \delta_2 \text{mpg}_i + \delta_4 \text{mpg}_i^2 \end{aligned}$$

- ◆ The **marginal effects of wgt<sub>i</sub> for domestic and foreign cars** in Model 3.1 are given respectively by:

$$\text{For domestic cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0)}{\partial \text{wgt}_i} = \beta_1 + 2\beta_3 \text{wgt}_i + \beta_5 \text{mpg}_i$$

$$\begin{aligned} \text{For foreign cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1)}{\partial \text{wgt}_i} \\ = \beta_1 + 2\beta_3 \text{wgt}_i + \beta_5 \text{mpg}_i + \delta_1 + 2\delta_3 \text{wgt}_i + \delta_5 \text{mpg}_i \end{aligned}$$

These two functions are identical (for any given values of wgt<sub>i</sub> and mpg<sub>i</sub>) if  $\delta_1 = 0$  and  $\delta_3 = 0$  and  $\delta_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$H_0: \delta_j = 0 \quad \text{for all } j = 1, 3, 5$$

$$\delta_1 = 0 \text{ and } \delta_3 = 0 \text{ and } \delta_5 = 0$$

$$H_1: \delta_j \neq 0 \quad j = 1, 3, 5$$

$$\delta_1 \neq 0 \text{ and/or } \delta_3 \neq 0 \text{ and/or } \delta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test fwgt fwgtsq fwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

- ◆ ***Test 4 - Model 3.1:*** Test the proposition that the ***foreign-domestic difference in mean car prices does not depend on the explanatory variable mpg<sub>i</sub>***.

- ◆ This proposition is empirically equivalent to the following three statements:

(1) The relationship of price<sub>i</sub> to mpg<sub>i</sub> is identical for domestic and foreign cars.

(2) The marginal effect of mpg<sub>i</sub> on price<sub>i</sub> is identical for domestic and foreign cars.

(3) The foreign-domestic difference in mean car prices is a function only of the explanatory variable wgt<sub>i</sub>.

- ◆ Recall that the foreign-domestic difference in conditional mean car prices for any specified values of wgt<sub>i</sub> and mpg<sub>i</sub> is given in Model 3.1 by

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0) \\ = \delta_0 + \delta_1 \text{wgt}_i + \delta_2 \text{mpg}_i + \delta_3 \text{wgt}_i^2 + \delta_4 \text{mpg}_i^2 + \delta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

The foreign-domestic difference in mean car prices does not depend on mpg<sub>i</sub> if  $\delta_2 = 0$  and  $\delta_4 = 0$  and  $\delta_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0) \\ = \delta_0 + \delta_1 \text{wgt}_i + \delta_3 \text{wgt}_i^2 \end{aligned}$$

- ◆ The **marginal effects of mpg<sub>i</sub> for domestic and foreign cars** in Model 3.1 are given respectively by:

$$\begin{aligned} \text{For domestic cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0)}{\partial \text{mpg}_i} \\ = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5 \text{wgt}_i \end{aligned}$$

$$\begin{aligned} \text{For foreign cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 1)}{\partial \text{mpg}_i} \\ = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5 \text{wgt}_i + \delta_2 + 2\delta_4 \text{mpg}_i + \delta_5 \text{wgt}_i \end{aligned}$$

These two functions are identical (for any given values of wgt<sub>i</sub> and mpg<sub>i</sub>) if  $\delta_2 = 0$  and  $\delta_4 = 0$  and  $\delta_5 = 0$ .

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$\begin{aligned} H_0: \delta_j = 0 \quad \text{for all } j = 2, 4, 5 \\ \delta_2 = 0 \text{ and } \delta_4 = 0 \text{ and } \delta_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \delta_j \neq 0 \quad j = 2, 4, 5 \\ \delta_2 \neq 0 \text{ and/or } \delta_4 \neq 0 \text{ and/or } \delta_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test fmpg fmpgsq fwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

- ◆ ***Test 5 - Model 3.1:*** Test the proposition that the *foreign-domestic difference in mean car prices is a linear function* of the explanatory variables  $wgt_i$  and  $mpg_i$ .
- ◆ Recall that the foreign-domestic difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.1 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, f_i = 1) - E(\text{price}_i | wgt_i, mpg_i, f_i = 0) \\ = \delta_0 + \delta_1 wgt_i + \delta_2 mpg_i + \delta_3 wgt_i^2 + \delta_4 mpg_i^2 + \delta_5 wgt_i mpg_i \end{aligned}$$

The foreign-domestic difference in mean car prices is a linear function of  $wgt_i$  and  $mpg_i$  if and only if  $\delta_3 = 0$  and  $\delta_4 = 0$  and  $\delta_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, f_i = 1) - E(\text{price}_i | wgt_i, mpg_i, f_i = 0) \\ = \delta_0 + \delta_1 wgt_i + \delta_2 mpg_i \end{aligned}$$

- ◆ Note the implications of the three coefficient restrictions  $\delta_3 = 0$ ,  $\delta_4 = 0$  and  $\delta_5 = 0$  for the **marginal effects of  $wgt_i$  and  $mpg_i$  for foreign cars** in Model 3.1, which are given respectively by:

$$\begin{aligned} \text{For } wgt_i: \quad & \frac{\partial E(\text{price}_i | wgt_i, mpg_i, f_i = 1)}{\partial wgt_i} \\ & = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i + \delta_1 + 2\delta_3 wgt_i + \delta_5 mpg_i \end{aligned}$$

$$\begin{aligned} \text{For } mpg_i: \quad & \frac{\partial E(\text{price}_i | wgt_i, mpg_i, f_i = 1)}{\partial mpg_i} \\ & = \beta_2 + 2\beta_4 mpg_i + \beta_5 wgt_i + \delta_2 + 2\delta_4 mpg_i + \delta_5 wgt_i \end{aligned}$$

Under the coefficient restrictions  $\delta_3 = 0$  and  $\delta_4 = 0$  and  $\delta_5 = 0$ , the marginal effects of  $wgt_i$  and  $mpg_i$  for *foreign cars* are:

$$\text{For } wgt_i: \quad \frac{\partial E(\text{price}_i | wgt_i, mpg_i, f_i = 1)}{\partial wgt_i} = \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i + \delta_1$$

$$\text{For } mpg_i: \quad \frac{\partial E(\text{price}_i | wgt_i, mpg_i, f_i = 1)}{\partial mpg_i} = \beta_2 + 2\beta_4 mpg_i + \beta_5 wgt_i + \delta_2$$

In other words, under the coefficient restrictions  $\delta_3 = 0$  and  $\delta_4 = 0$  and  $\delta_5 = 0$ , the marginal effects of  $\text{wgt}_i$  and  $\text{mpg}_i$  for *foreign cars* differ from the marginal effects of  $\text{wgt}_i$  and  $\text{mpg}_i$  for *domestic cars* only by a constant.

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$H_0: \delta_j = 0 \quad \text{for all } j = 3, 4, 5$$

$$\delta_3 = 0 \text{ and } \delta_4 = 0 \text{ and } \delta_5 = 0$$

$$H_1: \delta_j \neq 0 \quad j = 3, 4, 5$$

$$\delta_3 \neq 0 \text{ and/or } \delta_4 \neq 0 \text{ and/or } \delta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test fwgtsq fmpgsq fwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

### ***Model 3.1: Tests of the Regression Function for Domestic Cars***

- ◆ ***Test 6 - Model 3.1:*** Test the proposition that the **marginal effect of  $\text{wgt}_i$  is zero for domestic cars**.
- ◆ In equation (3.1), the marginal effect of  $\text{wgt}_i$  on  $\text{price}_i$  for domestic cars is:

$$\frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, f_i = 0)}{\partial \text{wgt}_i} = \beta_1 + 2\beta_3 \text{wgt}_i + \beta_5 \text{mpg}_i$$

This function equals zero for any given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  if  $\beta_1 = 0$  and  $\beta_3 = 0$  and  $\beta_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$H_0: \beta_j = 0 \quad \text{for all } j = 1, 3, 5$$

$$\beta_1 = 0 \text{ and } \beta_3 = 0 \text{ and } \beta_5 = 0$$

$$H_1: \beta_j \neq 0 \quad j = 1, 3, 5$$

$$\beta_1 \neq 0 \text{ and/or } \beta_3 \neq 0 \text{ and/or } \beta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test wgt wgtsq wgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

- ◆ **Test 7 - Model 3.1:** Test the proposition that the **marginal effect of  $mpg_i$  is zero for domestic cars.**
- ◆ In equation (3.1), the marginal effect of  $mpg_i$  on  $price_i$  for domestic cars is:

$$\frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, f_i = 0)}{\partial \text{mpg}_i} = \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5 \text{wgt}_i$$

This function equals zero for any given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  if  $\beta_2 = 0$  and  $\beta_4 = 0$  and  $\beta_5 = 0$ .

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$H_0: \beta_j = 0 \quad \text{for all } j = 2, 4, 5$$

$$\beta_2 = 0 \text{ and } \beta_4 = 0 \text{ and } \beta_5 = 0$$

$$H_1: \beta_j \neq 0 \quad j = 2, 4, 5$$

$$\beta_2 \neq 0 \text{ and/or } \beta_4 \neq 0 \text{ and/or } \beta_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test mpg mpgsq wgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis?

**Model 3.1: Tests of the Regression Function for Foreign Cars**

- ◆ **Test 8 - Model 3.1:** Test the proposition that the **marginal effect of  $wgt_i$  is zero for foreign cars.**
- ◆ In equation (3.1), the marginal effect of  $wgt_i$  on  $price_i$  for foreign cars is:

$$\begin{aligned} \frac{\partial E(\text{price}_i \mid wgt_i, mpg_i, f_i = 1)}{\partial wgt_i} &= \beta_1 + 2\beta_3 wgt_i + \beta_5 mpg_i + \delta_1 + 2\delta_3 wgt_i + \delta_5 mpg_i \\ &= (\beta_1 + \delta_1) + 2(\beta_3 + \delta_3)wgt_i + (\beta_5 + \delta_5)mpg_i \end{aligned}$$

This function equals zero for any given values of  $wgt_i$  and  $mpg_i$  if  $\beta_1 + \delta_1 = 0$  and  $\beta_3 + \delta_3 = 0$  and  $\beta_5 + \delta_5 = 0$ .

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$\begin{aligned} H_0: \beta_j + \delta_j = 0 & \quad \text{for all } j = 1, 3, 5 \\ \beta_1 + \delta_1 = 0 \text{ and } \beta_3 + \delta_3 = 0 \text{ and } \beta_5 + \delta_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \beta_j + \delta_j \neq 0 & \quad j = 1, 3, 5 \\ \beta_1 + \delta_1 \neq 0 \text{ and/or } \beta_3 + \delta_3 \neq 0 \text{ and/or } \beta_5 + \delta_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the **test** commands:

```
test wgt + fwgt = 0, notest
test wgtsq + fwgtsq = 0, notest accumulate
test wgtmpg + fwgtmpg = 0, accumulate
return list
```

Note the use of the **notest** and **accumulate** options on these **test** commands. The **notest** option suppresses printing or displaying the results of the intermediate commands in a sequence of **test** commands, when we are interested only in the results of the final **test** command in the sequence. The **accumulate** option is used to link a series of **test** commands in order to perform a **joint test** of all the

coefficient restrictions in the sequence, in this case the three coefficient restrictions specified by the null hypothesis  $H_0$ . Inspect the results of this series of three **test** commands. Would you reject or retain the null hypothesis  $H_0$ ?

- ◆ ***Test 9 - Model 3.1:*** Test the proposition that the **marginal effect of mpg<sub>i</sub> is zero for foreign cars**.
- ◆ In equation (3.1), the marginal effect of mpg<sub>i</sub> on price<sub>i</sub> for foreign cars is:

$$\begin{aligned} \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, f_i = 1)}{\partial \text{mpg}_i} &= \beta_2 + 2\beta_4 \text{mpg}_i + \beta_5 \text{wgt}_i + \delta_2 + 2\delta_4 \text{mpg}_i + \delta_5 \text{wgt}_i \\ &= (\beta_2 + \delta_2) + 2(\beta_4 + \delta_4) \text{mpg}_i + (\beta_5 + \delta_5) \text{wgt}_i \end{aligned}$$

This function equals zero for any given values of wgt<sub>i</sub> and mpg<sub>i</sub> if  $\beta_2 + \delta_2 = 0$  and  $\beta_4 + \delta_4 = 0$  and  $\beta_5 + \delta_5 = 0$ .

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$\begin{aligned} H_0: \beta_j + \delta_j = 0 & \quad \text{for all } j = 2, 4, 5 \\ \beta_2 + \delta_2 = 0 \text{ and } \beta_4 + \delta_4 = 0 \text{ and } \beta_5 + \delta_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \beta_j + \delta_j \neq 0 & \quad j = 2, 4, 5 \\ \beta_2 + \delta_2 \neq 0 \text{ and/or } \beta_4 + \delta_4 \neq 0 \text{ and/or } \beta_5 + \delta_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the **test** commands:

```
test mpg + fmpg = 0, notest
test mpgsq + fmpgsq = 0, accumulate notest
test wgtmpg + fwgtmpg = 0, accumulate
return list
```

Inspect the results of these commands. Would you reject or retain the null hypothesis  $H_0$ ?

**Model 3.2 – Version 2 of Model 3**

$$Y_i = \alpha_0 + \alpha_1 \text{wgt}_i + \alpha_2 \text{mpg}_i + \alpha_3 \text{wgt}_i^2 + \alpha_4 \text{mpg}_i^2 + \alpha_5 \text{wgt}_i \text{mpg}_i + \gamma_0 d_i + \gamma_1 d_i \text{wgt}_i + \gamma_2 d_i \text{mpg}_i + \gamma_3 d_i \text{wgt}_i^2 + \gamma_4 d_i \text{mpg}_i^2 + \gamma_5 d_i \text{wgt}_i \text{mpg}_i + u_i \quad (3.2)$$

- Before estimating Model 3.2, you need to create the interaction terms in the domestic-car indicator variable  $d_i$ , namely  $d_i \text{wgt}_i$ ,  $d_i \text{mpg}_i$ ,  $d_i \text{wgt}_i^2$ ,  $d_i \text{mpg}_i^2$  and  $d_i \text{wgt}_i \text{mpg}_i$ . Enter the following **generate** commands:

```
generate dwgt = d*wgt
generate dmpg = d*mpg
generate dwgtsq = d*wgtsq
generate dmpgsq = d*mpgsq
generate dwgtmpg = d*wgtmpg
```

- Estimate full-interaction regression equation (3.2) by OLS. Enter the **regress** command:

```
regress price wgt mpg wgtsq mpgsq wgtmpg d dwgt dmpg
dwgtsq dmpgsq dwgtmpg
```

- Use the **scalar** command to save the OLS coefficient estimates of equation (3.2), and the **scalar list** command to display the saved scalars. Enter the commands:

```
scalar alpha1 = _b[wgt]
scalar alpha2 = _b[mpg]
scalar alpha3 = _b[wgtsq]
scalar alpha4 = _b[mpgsq]
scalar alpha5 = _b[wgtmpg]
scalar alpha0 = _b[_cons]
scalar gamma1 = _b[dwgt]
scalar gamma2 = _b[dmpg]
scalar gamma3 = _b[dwgtsq]
scalar gamma4 = _b[dmpgsq]
scalar gamma5 = _b[dwgtmpg]
scalar gamma0 = _b[d]
scalar list alpha1 alpha2 alpha3 alpha4 alpha5 alpha0
scalar list gamma1 gamma2 gamma3 gamma4 gamma5 gamma0
```

- Use the **lincom** command to compute the *domestic-car coefficient estimates* implied by the OLS coefficient estimates of equation (3.2), and to perform two-tail significance tests of these coefficient estimates. Enter the commands:

```
lincom _b[wgt] + _b[dwgt]
lincom _b[mpg] + _b[dmpg]
lincom _b[wgtsq] + _b[dwgtsq]
lincom _b[mpgsq] + _b[dmpgsq]
lincom _b[wgtmpg] + _b[dwgtmpg]
lincom _b[_cons] + _b[d]
```

Which of the domestic-car coefficient estimates are individually statistically significant at the 1 percent significance level, at the 5 percent significance level, and at the 10 percent significance level?

### *Model 3.2: Tests for Coefficient Differences Between Foreign and Domestic Cars*

- ◆ **Test 1 - Model 3.2:** Test the proposition that *domestic cars and foreign cars have identical mean prices* for any given values of  $wgt_i$  and  $mpg_i$ .
- ◆ The domestic-foreign difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.2 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) \\ = \gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i \end{aligned}$$

- ◆ The proposition to be tested is that

$$E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) = 0 \quad \forall i$$

and hence that

$$\gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i = 0 \quad \text{for all } i$$

A **sufficient condition** for the above equation to be true is that **all six of the  $\gamma_j$  coefficients** in Model 3.2 jointly **equal zero**.

- ◆ The **null and alternative hypotheses** are as follows:

$$H_0: \gamma_j = 0 \quad \text{for all } j = 0, 1, \dots, 5$$

$$\gamma_0 = 0 \text{ and } \gamma_1 = 0 \text{ and } \gamma_2 = 0 \text{ and } \gamma_3 = 0 \text{ and } \gamma_4 = 0 \text{ and } \gamma_5 = 0$$

$$H_1: \gamma_j \neq 0 \quad j = 0, 1, \dots, 5$$

$$\gamma_0 \neq 0 \text{ and/or } \gamma_1 \neq 0 \text{ and/or } \gamma_2 \neq 0 \text{ and/or } \gamma_3 \neq 0 \text{ and/or } \gamma_4 \neq 0 \text{ and/or } \gamma_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test d dwgt dmpg dwgtsq dmpgsq dwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 1 for Model 3.1. Are they identical tests?

- ◆ **Test 2 - Model 3.2:** Test the proposition that the **domestic-foreign difference in mean car prices is a constant**, i.e., that it does not depend on the explanatory variables  $wgt_i$  and  $mpg_i$ .
- ◆ Recall that the domestic-foreign difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.2 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) \\ = \gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i \end{aligned}$$

- ◆ The proposition to be tested is that

$$\gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i = \gamma_0 \quad \text{for all } i$$

A **sufficient condition** for the above equation to be true is that **the five  $\gamma_j$  coefficients** on the domestic dummy variable interaction terms in Model 3.2 all **equal zero**.

- ♦ The **null and alternative hypotheses** are as follows:

$$H_0: \gamma_j = 0 \quad \text{for all } j = 1, \dots, 5$$

$$\gamma_1 = 0 \text{ and } \gamma_2 = 0 \text{ and } \gamma_3 = 0 \text{ and } \gamma_4 = 0 \text{ and } \gamma_5 = 0$$

$$H_1: \gamma_j \neq 0 \quad j = 1, \dots, 5$$

$$\gamma_1 \neq 0 \text{ and/or } \gamma_2 \neq 0 \text{ and/or } \gamma_3 \neq 0 \text{ and/or } \gamma_4 \neq 0 \text{ and/or } \gamma_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test dwgt dmpg dwgtsq dmpgsq dwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 2 for Model 3.1. Are they identical tests?

- ♦ ***Test 3 - Model 3.2:*** Test the proposition that the **domestic-foreign difference in mean car prices does not depend on the explanatory variable  $wgt_i$ .**

- ♦ This proposition is empirically equivalent to the following three statements:

(1) The relationship of  $price_i$  to  $wgt_i$  is identical for domestic and foreign cars.

(2) The marginal effect of  $wgt_i$  on  $price_i$  is identical for domestic and foreign cars.

(3) The domestic-foreign difference in mean car prices is a function only of the explanatory variable  $mpg_i$ .

- ♦ Recall that the domestic-foreign difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.2 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) \\ = \gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i \end{aligned}$$

The domestic-foreign difference in mean car prices does not depend on  $wgt_i$  if and only if  $\gamma_1 = 0$  and  $\gamma_3 = 0$  and  $\gamma_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 0) \\ = \gamma_0 + \gamma_2 \text{mpg}_i + \gamma_4 \text{mpg}_i^2 \end{aligned}$$

- ◆ The **marginal effects of  $\text{wgt}_i$  for *domestic and foreign cars*** in Model 3.2 are given respectively by:

$$\begin{aligned} \text{For domestic cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{wgt}_i} \\ = \alpha_1 + 2\alpha_3 \text{wgt}_i + \alpha_5 \text{mpg}_i + \gamma_1 + 2\gamma_3 \text{wgt}_i + \gamma_5 \text{mpg}_i \end{aligned}$$

$$\begin{aligned} \text{For foreign cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 0)}{\partial \text{wgt}_i} \\ = \alpha_1 + 2\alpha_3 \text{wgt}_i + \alpha_5 \text{mpg}_i \end{aligned}$$

These two functions are identical for any given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  if  $\gamma_1 = 0$  and  $\gamma_3 = 0$  and  $\gamma_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$\begin{aligned} H_0: \gamma_j = 0 \quad \text{for all } j = 1, 3, 5 \\ \gamma_1 = 0 \text{ and } \gamma_3 = 0 \text{ and } \gamma_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \gamma_j \neq 0 \quad j = 1, 3, 5 \\ \gamma_1 \neq 0 \text{ and/or } \gamma_3 \neq 0 \text{ and/or } \gamma_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test dwgt dwgtsq dwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 3 for Model 3.1. Are they identical tests?

- ◆ **Test 4 - Model 3.2:** Test the proposition that the *domestic-foreign difference in mean car prices does not depend on the explanatory variable mpg<sub>i</sub>*.
- ◆ This proposition is empirically equivalent to the following three statements:
  - (1) The relationship of price<sub>i</sub> to mpg<sub>i</sub> is identical for domestic and foreign cars.
  - (2) The marginal effect of mpg<sub>i</sub> on price<sub>i</sub> is identical for domestic and foreign cars.
  - (3) The domestic-foreign difference in mean car prices is a function only of the explanatory variable wgt<sub>i</sub>.
- ◆ Recall that the domestic-foreign difference in conditional mean car prices for any specified values of wgt<sub>i</sub> and mpg<sub>i</sub> is given in Model 3.2 by

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 0) \\ = \gamma_0 + \gamma_1 \text{wgt}_i + \gamma_2 \text{mpg}_i + \gamma_3 \text{wgt}_i^2 + \gamma_4 \text{mpg}_i^2 + \gamma_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

The domestic-foreign difference in mean car prices does not depend on mpg<sub>i</sub> if and only if  $\gamma_2 = 0$  and  $\gamma_4 = 0$  and  $\gamma_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 1) - E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 0) \\ = \gamma_0 + \gamma_1 \text{wgt}_i + \gamma_3 \text{wgt}_i^2 \end{aligned}$$

- ◆ The **marginal effects of mpg<sub>i</sub> for domestic and foreign cars** in Model 3.2 are given respectively by:

$$\begin{aligned} \text{For domestic cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{mpg}_i} \\ = \alpha_2 + 2\alpha_4 \text{mpg}_i + \alpha_5 \text{wgt}_i + \gamma_2 + 2\gamma_4 \text{mpg}_i + \gamma_5 \text{wgt}_i \end{aligned}$$

$$\begin{aligned} \text{For foreign cars: } \frac{\partial E(\text{price}_i | \text{wgt}_i, \text{mpg}_i, d_i = 0)}{\partial \text{mpg}_i} \\ = \alpha_2 + 2\alpha_4 \text{mpg}_i + \alpha_5 \text{wgt}_i \end{aligned}$$

These two functions are identical for any given values of  $wgt_i$  and  $mpg_i$  if  $\gamma_2 = 0$  and  $\gamma_4 = 0$  and  $\gamma_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$H_0: \gamma_j = 0 \quad \text{for all } j = 2, 4, 5$$

$$\gamma_2 = 0 \text{ and } \gamma_4 = 0 \text{ and } \gamma_5 = 0$$

$$H_1: \gamma_j \neq 0 \quad j = 2, 4, 5$$

$$\gamma_2 \neq 0 \text{ and/or } \gamma_4 \neq 0 \text{ and/or } \gamma_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test dmpg dmpgsq dwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 4 for Model 3.1. Are they identical tests?

- ◆ ***Test 5 - Model 3.2:*** Test the proposition that the ***domestic-foreign difference in mean car prices is a linear function*** of the explanatory variables  $wgt_i$  and  $mpg_i$ .
- ◆ Recall that the domestic-foreign difference in conditional mean car prices for any specified values of  $wgt_i$  and  $mpg_i$  is given in Model 3.2 by

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) \\ = \gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i + \gamma_3 wgt_i^2 + \gamma_4 mpg_i^2 + \gamma_5 wgt_i mpg_i \end{aligned}$$

The domestic-foreign difference in mean car prices is a linear function of  $wgt_i$  and  $mpg_i$  if and only if  $\gamma_3 = 0$  and  $\gamma_4 = 0$  and  $\gamma_5 = 0$ . Under these three exclusion restrictions,

$$\begin{aligned} E(\text{price}_i | wgt_i, mpg_i, d_i = 1) - E(\text{price}_i | wgt_i, mpg_i, d_i = 0) \\ = \gamma_0 + \gamma_1 wgt_i + \gamma_2 mpg_i \end{aligned}$$

- ◆ Note the implications of the three coefficient restrictions  $\gamma_3 = 0$ ,  $\gamma_4 = 0$  and  $\gamma_5 = 0$  for the **marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> for domestic cars** in Model 3.2, which are given respectively by:

$$\begin{aligned} \text{For wgt}_i: \quad & \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{wgt}_i} \\ & = \alpha_1 + 2\alpha_3 \text{wgt}_i + \alpha_5 \text{mpg}_i + \gamma_1 + 2\gamma_3 \text{wgt}_i + \gamma_5 \text{mpg}_i \end{aligned}$$

$$\begin{aligned} \text{For mpg}_i: \quad & \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{mpg}_i} \\ & = \alpha_2 + 2\alpha_4 \text{mpg}_i + \alpha_5 \text{wgt}_i + \gamma_2 + 2\gamma_4 \text{mpg}_i + \gamma_5 \text{wgt}_i \end{aligned}$$

Under the coefficient restrictions  $\gamma_3 = 0$  and  $\gamma_4 = 0$  and  $\gamma_5 = 0$ , the marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> for *domestic cars* are:

$$\text{For wgt}_i: \quad \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{wgt}_i} = \alpha_1 + 2\alpha_3 \text{wgt}_i + \alpha_5 \text{mpg}_i + \gamma_1$$

$$\text{For mpg}_i: \quad \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{mpg}_i} = \alpha_2 + 2\alpha_4 \text{mpg}_i + \alpha_5 \text{wgt}_i + \gamma_2$$

In other words, under the coefficient restrictions  $\gamma_3 = 0$  and  $\gamma_4 = 0$  and  $\gamma_5 = 0$ , the marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> for *domestic cars* differ from the marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> for *foreign cars* only by a constant.

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$H_0: \gamma_j = 0 \quad \text{for all } j = 3, 4, 5$$

$$\gamma_3 = 0 \text{ and } \gamma_4 = 0 \text{ and } \gamma_5 = 0$$

$$H_1: \gamma_j \neq 0 \quad j = 3, 4, 5$$

$$\gamma_3 \neq 0 \text{ and/or } \gamma_4 \neq 0 \text{ and/or } \gamma_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test dwgtsq dmpgsq dwgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 5 for Model 3.1. Are they identical tests?

### *Model 3.2: Tests of the Regression Function for Domestic Cars*

- ◆ ***Test 6 - Model 3.2:*** Test the proposition that the **marginal effect of  $wgt_i$  is zero for domestic cars.**
- ◆ In equation (3.2), the marginal effect of  $wgt_i$  on  $price_i$  for domestic cars is:

$$\begin{aligned} \frac{\partial E(\text{price}_i \mid wgt_i, mpg_i, d_i = 1)}{\partial wgt_i} \\ &= \alpha_1 + 2\alpha_3 wgt_i + \alpha_5 mpg_i + \gamma_1 + 2\gamma_3 wgt_i + \gamma_5 mpg_i \\ &= (\alpha_1 + \gamma_1) + 2(\alpha_3 + \gamma_3)wgt_i + (\alpha_5 + \gamma_5)mpg_i \end{aligned}$$

This function equals zero for any given values of  $wgt_i$  and  $mpg_i$  if  $\alpha_1 + \gamma_1 = 0$  and  $\alpha_3 + \gamma_3 = 0$  and  $\alpha_5 + \gamma_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$\begin{aligned} H_0: \alpha_j + \gamma_j = 0 \quad \text{for all } j = 1, 3, 5 \\ \alpha_1 + \gamma_1 = 0 \text{ and } \alpha_3 + \gamma_3 = 0 \text{ and } \alpha_5 + \gamma_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \alpha_j + \gamma_j \neq 0 \quad j = 1, 3, 5 \\ \alpha_1 + \gamma_1 \neq 0 \text{ and/or } \alpha_3 + \gamma_3 \neq 0 \text{ and/or } \alpha_5 + \gamma_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the **test** commands:

```
test wgt + dwgt = 0, notest
test wgtsq + dwgtsq = 0, accumulate notest
test wgtmpg + dwgtmpg = 0, accumulate
return list
```

Inspect the results of these **test** commands. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 6 for Model 3.1. Are they identical tests?

- ◆ ***Test 7 - Model 3.2:*** Test the proposition that the **marginal effect of mpg<sub>i</sub> is zero for domestic cars.**

- ◆ In equation (3.2), the marginal effect of mpg<sub>i</sub> on price<sub>i</sub> for domestic cars is:

$$\begin{aligned} \frac{\partial E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1)}{\partial \text{mpg}_i} \\ &= \alpha_2 + 2\alpha_4 \text{mpg}_i + \alpha_5 \text{wgt}_i + \gamma_2 + 2\gamma_4 \text{mpg}_i + \gamma_5 \text{wgt}_i \\ &= (\alpha_2 + \gamma_2) + 2(\alpha_4 + \gamma_4) \text{mpg}_i + (\alpha_5 + \gamma_5) \text{wgt}_i \end{aligned}$$

This function equals zero for any given values of wgt<sub>i</sub> and mpg<sub>i</sub> if  $\alpha_2 + \gamma_2 = 0$  and  $\alpha_4 + \gamma_4 = 0$  and  $\alpha_5 + \gamma_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$\begin{aligned} H_0: \alpha_j + \gamma_j = 0 \quad \text{for all } j = 2, 4, 5 \\ \alpha_2 + \gamma_2 = 0 \text{ and } \alpha_4 + \gamma_4 = 0 \text{ and } \alpha_5 + \gamma_5 = 0 \end{aligned}$$

$$\begin{aligned} H_1: \alpha_j + \gamma_j \neq 0 \quad j = 2, 4, 5 \\ \alpha_2 + \gamma_2 \neq 0 \text{ and/or } \alpha_4 + \gamma_4 \neq 0 \text{ and/or } \alpha_5 + \gamma_5 \neq 0 \end{aligned}$$

- Perform a Wald F-test of the above hypothesis. Enter the **test** commands:

```
test mpg + dmpg = 0, notest
test mpgsq + dmpgsq = 0, accumulate notest
test wgtmpg + dwgtmpg = 0, accumulate
return list
```

Inspect the results of these **test** commands. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 7 for Model 3.1. Are they identical tests?

**Model 3.2: Tests of the Regression Function for Foreign Cars**

- ◆ **Test 8 - Model 3.2:** Test the proposition that the **marginal effect of  $wgt_i$  is zero for foreign cars.**

- ◆ In equation (3.2), the marginal effect of  $wgt_i$  on  $price_i$  for foreign cars is:

$$\frac{\partial E(\text{price}_i \mid wgt_i, mpg_i, d_i = 0)}{\partial wgt_i} = \alpha_1 + 2\alpha_3 wgt_i + \alpha_5 mpg_i$$

This function equals zero for any given values of  $wgt_i$  and  $mpg_i$  if  $\alpha_1 = 0$  and  $\alpha_3 = 0$  and  $\alpha_5 = 0$ .

- ◆ The **null and alternative hypotheses** are therefore as follows:

$$H_0: \alpha_j = 0 \quad \text{for all } j = 1, 3, 5$$

$$\alpha_1 = 0 \text{ and } \alpha_3 = 0 \text{ and } \alpha_5 = 0$$

$$H_1: \alpha_j \neq 0 \quad j = 1, 3, 5$$

$$\alpha_1 \neq 0 \text{ and/or } \alpha_3 \neq 0 \text{ and/or } \alpha_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test wgt wgtsq wgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 8 for Model 3.1. Are they identical tests?

- ◆ **Test 9 - Model 3.2:** Test the proposition that the **marginal effect of  $mpg_i$  is zero for foreign cars.**

- ◆ In equation (3.2), the marginal effect of  $mpg_i$  on  $price_i$  for foreign cars is:

$$\frac{\partial E(\text{price}_i \mid wgt_i, mpg_i, d_i = 0)}{\partial mpg_i} = \alpha_2 + 2\alpha_4 mpg_i + \alpha_5 wgt_i$$

This function equals zero for any given values of  $wgt_i$  and  $mpg_i$  if  $\alpha_2 = 0$  and  $\alpha_4 = 0$  and  $\alpha_5 = 0$ .

- ◆ The ***null and alternative hypotheses*** are therefore as follows:

$$H_0: \alpha_j = 0 \quad \text{for all } j = 2, 4, 5$$

$$\alpha_2 = 0 \text{ and } \alpha_4 = 0 \text{ and } \alpha_5 = 0$$

$$H_1: \alpha_j \neq 0 \quad j = 2, 4, 5$$

$$\alpha_2 \neq 0 \text{ and/or } \alpha_4 \neq 0 \text{ and/or } \alpha_5 \neq 0$$

- Perform a Wald F-test of the above hypothesis. Enter the commands:

```
test mpg mpgsq wgtmpg
return list
```

Inspect the results of this **test** command. Would you reject or retain the null hypothesis? Compare the results of this test with those of Test 9 for Model 3.1. Are they identical tests?

### **Comparing Regression Models 3.1 and 3.2: Summary**

The estimates and hypothesis tests you have performed on Models 3.1 and 3.2 demonstrate that these two regression models are ***observationally equivalent*** – that is, they represent two different but equivalent ways of estimating two separate price<sub>i</sub> equations for domestic and foreign cars sold in North America, of testing hypotheses about how the price<sub>i</sub> equations for domestic and foreign cars differ from one another, and of testing hypotheses about the marginal effects of the explanatory variables  $wgt_i$  and  $mpg_i$  on price<sub>i</sub> for domestic and foreign cars.

- ◆ The equivalence of Models 3.1 and 3.2 is demonstrated by the following results:
  1. Models 3.1 and 3.2 yield *identical* coefficient estimates of the price<sub>i</sub> regression function for ***domestic cars***, i.e.,

$$\hat{\beta}_j \text{ in Model 3.1} = \hat{\alpha}_j + \hat{\gamma}_j \text{ in Model 3.2,} \quad j = 0, 1, \dots, 5.$$

2. Models 3.1 and 3.2 yield *identical* coefficient estimates of the price<sub>i</sub> regression function for *foreign cars*, i.e.,

$$\hat{\alpha}_j \text{ in Model 3.2} = \hat{\beta}_j + \hat{\delta}_j \text{ in Model 3.1,} \quad j = 0, 1, \dots, 5.$$

- ♦ The equivalence of the tests performed on Models 3.1 and 3.2 for *coefficient differences* between the regression functions for domestic and foreign cars is demonstrated by the fact that Tests 1 to 5 on Model 3.1 yield identical results to Tests 1 to 5 on Model 3.2.
- ♦ The equivalence of the tests performed on Models 3.1 and 3.2 respecting the marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> on the mean prices of *domestic cars* is demonstrated by the finding that Tests 6 and 7 on Model 3.1 yield identical results to Tests 6 and 7 on Model 3.2.
- ♦ The equivalence of the tests performed on Models 3.1 and 3.2 respecting the marginal effects of wgt<sub>i</sub> and mpg<sub>i</sub> on the mean prices of *foreign cars* is demonstrated by the finding that Tests 8 and 9 on Model 3.1 yield identical results to Tests 8 and 9 on Model 3.2.

### Model 3.3 – Version 3 of Model 3: No Base Group

This may come as unwelcome news at this point, but there is yet another specification of the full-interaction regression model, Model 3. This one does not use either domestic cars or foreign cars as a base group. Its specification is as follows:

$$Y_i = \beta_0 d_i + \beta_1 d_i \text{wgt}_i + \beta_2 d_i \text{mpg}_i + \beta_3 d_i \text{wgt}_i^2 + \beta_4 d_i \text{mpg}_i^2 + \beta_5 d_i \text{wgt}_i \text{mpg}_i + \alpha_0 f_i + \alpha_1 f_i \text{wgt}_i + \alpha_2 f_i \text{mpg}_i + \alpha_3 f_i \text{wgt}_i^2 + \alpha_4 f_i \text{mpg}_i^2 + \alpha_5 f_i \text{wgt}_i \text{mpg}_i + u_i \quad (3.3)$$

The distinguishing feature of equation (3.3) is that it contains **no intercept coefficient**. It includes *both* dummy variables  $d_i$  and  $f_i$  as additive regressors, and a full set of dummy variable interaction terms in both  $d_i$  and  $f_i$ .

***Interpreting the Coefficients in Model 3.3***

- ◆ The **regression coefficients**  $\beta_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.3) are the regression coefficients for ***domestic cars***.

$\beta_j =$  ***domestic-car*** regression coefficient for regressor  $j$ .

To see why, use equation (3.3) to take the conditional mean of  $\text{price}_i$  for given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  when  $d_i = 1$  and  $f_i = 0$ ; this gives you the **regression function for *domestic cars***:

$$\begin{aligned} E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, d_i = 1, f_i = 0) \\ = \beta_0 + \beta_1 \text{wgt}_i + \beta_2 \text{mpg}_i + \beta_3 \text{wgt}_i^2 + \beta_4 \text{mpg}_i^2 + \beta_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

- ◆ The **regression coefficients**  $\alpha_j$  ( $j = 0, 1, \dots, 5$ ) in equation (3.3) are the regression coefficients for ***foreign cars***.

$\alpha_j =$  ***foreign-car*** regression coefficient for regressor  $j$ .

To see why, use equation (3.3) to take the conditional mean of  $\text{price}_i$  for given values of  $\text{wgt}_i$  and  $\text{mpg}_i$  when  $f_i = 1$  and  $d_i = 0$ ; this gives you the **regression function for *foreign cars***:

$$\begin{aligned} E(\text{price}_i \mid \text{wgt}_i, \text{mpg}_i, f_i = 1, d_i = 0) \\ = \alpha_0 + \alpha_1 \text{wgt}_i + \alpha_2 \text{mpg}_i + \alpha_3 \text{wgt}_i^2 + \alpha_4 \text{mpg}_i^2 + \alpha_5 \text{wgt}_i \text{mpg}_i \end{aligned}$$

- Estimate full-interaction regression equation (3.3) by OLS. Enter the following **regress** command with the **noconstant** option:

```
regress price d dwgt dmpg dwgtsq dmpgsq dwgtmpg f fwgt fmpg
fwgtsq fmpgsq fwgtmpg, noconstant
```

- Compare the OLS coefficient estimates of Model 3.3 with the separate OLS coefficient estimates for the domestic-car and foreign-car regression functions you have previously computed. Enter the following **scalar list** commands:

```
scalar list b0 b1 b2 b3 b4 b5
scalar list a0 a1 a2 a3 a4 a5
```

- Estimate by OLS (again) the **regression equation for *domestic cars*** on the subsample of observations for domestic cars. Enter *either* of the following **regress** commands with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if d==1
regress price wgt mpg wgtsq mpgsq wgtmpg if f==0
```

Compare the OLS coefficient estimates for domestic cars from these **regress** commands with those given by OLS estimation of Model 3.3.

- Estimate by OLS (again) the **regression equation for *foreign cars*** on the subsample of observations for foreign cars. Enter *either* of the following **regress** commands with the **if** option:

```
regress price wgt mpg wgtsq mpgsq wgtmpg if d==0
regress price wgt mpg wgtsq mpgsq wgtmpg if f==1
```

Compare the OLS coefficient estimates for foreign cars from these **regress** commands with those given by OLS estimation of Model 3.3.

### ***Optional Exercise on Model 3.3***

If you have the patience and perseverance for it, you may wish to formulate and perform Tests 1 to 9 on Model 3.3. By doing so, you will demonstrate for yourself that Model 3.3 is ***observationally equivalent*** to Models 3.1 and 3.2.

---

## □ **Preparing to End Your *Stata* Session**

---

**Before you end your *Stata* session**, you should do two things.

- First, you will want to **save the current dataset**. To save the current data set as *Stata*-format data set **auto5.dta**, enter either of the following **save** commands:

```
save auto5      or      save auto5, replace
```

- Second, **close the log file** you have been recording. Enter the command:

```
log close
```

- Finally, **close the command log file** you have been recording. Enter the command:

```
cmdlog close
```

---

## □ **End Your *Stata* Session -- exit**

---

- **To end your *Stata* session**, use the **exit** command. Enter the command:

```
exit      or      exit, clear
```

---

## □ **Cleaning Up and Clearing Out**

---

**After returning to Windows**, you should copy all the files you have used and created during your *Stata* session to your own portable electronic storage device such as a flash memory stick. These files will be found in the ***Stata working directory***, which is usually **C:\data** on the computers in Dunning 350. There are three files you will want to be sure you have: the complete *Stata* log file **452tutorial5.log**; the *Stata* command log file **452tutorial5.txt**; and the *Stata*-format dataset **auto5.dta**. Use the Windows **copy** command to copy any files you want to keep to your own portable electronic storage device (e.g., a flash memory stick).

Finally, **as a courtesy to other users** of the computing classroom, please delete all the files you have used or created from the *Stata* working directory.